
Corso di Laurea in Informatica 
Corso di Intelligenza Artificiale (prof. Nicola Leone) 
 

Prova d’esame del 09/01/2007 
 

 1/5
87036 Arcavacata di Rende (Cs) - Edificio  Polifunzionale   tel. (+ 39)  0984  493091  fax (+ 39)  0984  401186
CAMPUS DI ARCAVACATA www.unical.it

UNIVERSITÀ CALABRIADELLA

s rq

t 
- 

Esercizio 1.  
-------------- 
Svolgere tutti i punti. 
 
a) Dato il seguente programma Datalog, disegnarne il grafo delle dipendenze. Determinare poi l’unico 
answer set mediante il calcolo del minimo punto fisso dell’operatore TP. 
 
r(2). 
q(2). 
 
r(Z) :- s(X), Z=X+1, Z<4. 
s(X) :- r(X), not t(X). 
 
b) Si aggiunga al precedente programma la seguente regola:  
 
p(X) v t(X) :- q(X), not s(X).  
 
Calcolare quindi TUTTI gli answer set del programma risultante. 
 
c) Si aggiunga al precedente programma il seguente weak constraint. 
  
:~ r(X), s(Y). [X:Y] 
 
Calcolare quindi gli answer set riportando per ciascuno il costo. Indicare quindi quello ottimo. 
 
d) Si aggiunga il seguente strong constraint. 
 
:- r(X),t(X). 
 
Come influisce sulle soluzione del programma? Perché?   
 
------------------------ 
SOLUZIONE 
 
a)  
Il grafo delle dipendenze è riportato qui a lato.  
Il punto fisso dell’operatore Tp è: {q(2), r(2), r(3), s(2), s(3)}. 
-- 
 
b) c) 
% EDB facts: 
q(2). 
% Facts derived to be true in every answer set: 
r(2). 
% Residual ground instantiation of the program: 

p(2) v t(2) :- not s(2). 
s(2) :- not t(2). 
r(3) :- s(2). 
s(3) :- r(3). 
 
MODELS:  
{q(2), r(2), s(2), r(3), s(3)} 
Cost ([Weight:Level]): <[0:1],[5:2],[5:3]> 
 
{q(2), r(2), t(2)} 
Cost ([Weight:Level]): <[0:1],[0:2],[0:3]> 
 
OTTIMO: 
Best model: {q(2), r(2), t(2)} 
Cost ([Weight:Level]): <[0:1],[0:2],[0:3]> 
 
d)  
CON LO STRONG CONSTRAINT: 
Best model: {q(2), r(2), s(2), r(3), s(3)} 
Cost ([Weight:Level]): <[0:1],[5:2],[5:3]> 
 
 
Esercizio 2. 
--------------  
Ciccio Pasticcio ha avviato una nuova attività, un'agenzia matrimoniale dal nome "Cuori Disgiunti". I suoi 
affari però non vanno proprio bene. Ciccio è un gran pasticcione e finora ha creato coppie davvero male 
assortite: la ventenne con il sessantenne, la signora a cui piace la montagna con il signore che ama solo il 
mare, etc. Vogliamo aiutare Ciccio nella sua attività? 
Si scriva un programma DLV che crei delle coppie tra gli iscritti all'agenzia in modo da evitare 
assolutamente coppie con interessi contrastanti.  
 
Si tenga inoltre conto delle seguenti considerazioni: 
• Per ottenere il maggior profitto possibile, si vuole massimizzare il numero di coppie formate.   
• A parità di numero di coppie formate, si preferisce riuscire a formare coppie che abbiano il maggior 

numero possibile di cose in comune (ma si ricordi: solo a parità di numero di coppie formate). 
• Si vuole minimizzare il più possibile la differenza di età tra i componenti di ciascuna coppia; questa 

necessità è però meno importante delle altre. 
 
* Modello dei dati in input:  

iscritto(Nome,Età,Sesso).    gli iscritti all'agenzia 
 ama(Nome,Cosa).      cosa piace agli iscritti 
 odia(Nome,Cosa).      cosa gli iscritti odiano 
 
------------------------ 

SOLUZIONE 
 
iscritto(giovanni,57,m). 
iscritto(sonia,46,f). 
iscritto(andrea,29,m). 
iscritto(caterina,32,f). 
iscritto(giuseppina,30,f). 
iscritto(vittorio,47,m). 
 
ama(giovanni,musica). 
ama(sonia,mare). 
ama(caterina,feste). 
ama(andrea,feste). 
 
odia(giovanni,mare). 
 
coppia(M,F) v noncoppia(M,F) :- iscritto(M,_,m),iscritto(F,_,f). 
 
:- coppia(M1,F), coppia(M2,F), M1!=M2. 
:- coppia(M,F1), coppia(M,F2), F1!=F2. 
 
:- coppia(M,F), odia(M,C), ama(F,C). 
:- coppia(M,F), odia(F,C), ama(M,C). 
 
differenzaeta(M,F,D) :- iscritto(M,E1,_),iscritto(F,E2,_),E1=E2+D. 
differenzaeta(M,F,D) :- iscritto(M,E1,_),iscritto(F,E2,_),E2=E1+D. 
:~ coppia(M,F), differenzaeta(M,F,D). [D:1] 
 
:~ coppia(M,F), #count{C:ama(M,C), not ama(F,C)}=X,  
        #count{C: ama(F,C), not ama(M,C)}=Y, N=X+Y, 
        #int(X),#int(Y). [N:2]  
 
:~ noncoppia(M,F). [1:3] 
 
 
 
 
Esercizio 3.  
-------------- 
Ieri mattina il fattore Jake ha trovato al suo risveglio una bella  sorpresa nel cortile della sua fattoria. Il 
cancello era stato lasciato aperto e gli animali, durante la notte, se ne erano tutti (comprese le mucche) 
andati in giro. Dopo un'ora di ricerche Jake finalmente ha riportato tutti i suoi animali nei loro recinti. 
Utilizzando i seguenti indizi, per ogni specie di animali, determinare: 
- il numero di capi posseduti (2,4,5,6,12); 
- la situazione in cui jake li ha ritrovati (nell'orto, nel granaio, nel fienile, nell'aia, nel giardino del vicino); 

- il tempo impiegato per il recupero (5,10,11,15,20 minuti). 
 
1) Gli animali che gironzolavano nel giardino del vicino non erano le capre. 
 
2) Le dodici galline non stavano mangiando i suoi ortaggi. 
 
3) La specie ritrovata nel granaio è quella di cui Jake possiede il minor numero di capi. Per recuperare la 
specie di cui, invece, possiede il maggior numero di capi,  Jake ha impiegato venti minuti.   
 
4) Jake ha impiegato cinque minuti per recuperare la specie di cui possiede solo due capi, mentre ha 
impiegato il doppio del tempo per quella di cui possiede sei capi. 
 
5) Il numero di galline supera di sei unità il numero di capre che, a sua volta, supera di due il numero di 
pecore.  
 
6) Per recuperare i cavalli ci sono voluti cinque minuti in più del tempo impiegato a recuperare gli animali 
ritrovati nel fienile ma cinque minuti in meno del tempo impiegato per recuperare gli animali ritrovati 
nell'aia. 
 
7) Per recuperare le pecore Jake ha impiegato un minuto in più del tempo usato per recuperare gli animali 
ritrovati nel fienile. 
 
8) Il numero di animali ritrovati nell'orto e' multiplo del numero di animali ritrovati nel granaio ma il 
tempo impiegato per recuperarli non è multiplo del tempo impiegato per recuperare gli animali dal 
granaio. 
 
------------------------ 
SOLUZIONE 
 
specie(gallina). 
specie(pecora). 
specie(cavallo). 
specie(capra). 
specie(mucca). 
numero(2,S) v numero(4,S) v numero(5,S) v numero(6,S) v numero(12,S) :- specie(S).  
luogo(orto,S) v luogo(fienile,S) v luogo(granaio,S) v luogo(aia,S) v luogo(giardino,S) :- specie(S). 
tempo(5,S) v tempo(10,S) v tempo(11,S) v tempo(15,S) v tempo(20,S) :- specie(S). 
 
:- numero(N,S1), numero(N,S2), S1 != S2. 
:- luogo(L,S1), luogo(L,S2), S1 != S2. 
:- tempo(T,S1), tempo(T,S2), S1 != S2. 
 
:- luogo(giardino,capra). 
 
numero(12,gallina). 

:- luogo(orto,gallina). 
 
:- luogo(granaio,S), not numero(2,S). 
tempo(20,gallina). 
 
:- tempo(5,S), not numero(2,S). 
:- tempo(10,S), not numero(6,S). 
 
numero(6,capra). 
numero(4,pecora). 
 
cinqueminutidopo(T1,T2) :- T1 = T2+5. 
cinqueminutiprima(T1,T2) :- T2 = T1+5. 
:- tempo(T1,cavallo), tempo(T2,S), luogo(fienile,S), not cinqueminutidopo(T1,T2).  
:- tempo(T1,cavallo), tempo(T2,S), luogo(aia,S), not cinqueminutiprima(T1,T2).  
 
unminutodopo(T1,T2) :- T1 = T2+1. 
:- tempo(T1,pecora), tempo(T2,S), luogo(fienile,S), not unminutodopo(T1,T2).  
 
multiplo(N1,N2) :- N1=N2*X. 
:- numero(N1,S1), luogo(orto,S1), numero(N2,S2), luogo(granaio,S2), not multiplo(N1,N2). 
:- tempo(T1,S1), luogo(orto,S1), tempo(T2,S2), luogo(granaio,S2), multiplo(T1,T2). 
 


